

Puhelinyhteyksien komentokieli

Numeronvalitsinta varten

Copyright (c) 1997 Microsoft Corp.
Osa tästä materiaalista on Shiva Corporationin toimittamaa.

Tämän tiedoston käyttäminen

Tämä tiedosto on tarkoitettu pääasiassa Internet-palvelutoimittajille. Tiedostossa on ohjeita sellaisten komentosarjatiedostojen kirjoittamisesta, joita käyttäjät voivat muokata omien yhteystietojensa mukaisiksi.

Komentokielen käyttäminen Internet Explorerissa

IEscript.exe-ohjelmalla käyttäjä voi määrittää yhteyden komentosarjatiedoston (.scp) avulla. Iescript-ohjelma lisää määrytykset yhteystiedostoon (.con) seuraavassa osoitetulla tavalla. Muokkaa yhteystiedostoa aina Iescript-ohjelman avulla, älä manuaalisesti.

Osan nimi: **Script**

Seuraavia avainsanoja tuetaan:

ScriptEnabled

Arvo voi olla Yes tai No. Määrittää, aktivoidaanko komentokielityökalu.

ScriptFileName

Nimeämiskäytäntöjen mukainen merkkijono, joka määrittää tähän yhteystiedostoon liitetyn komentosarjatiedoston sijainnin.

ScriptTerminal

Arvo voi olla Yes tai No. Määrittää, näytetäänkö komentokielen päätteikkuna komennon suorittamisen aikana.

ScriptRecord

Arvo voi olla Yes tai No. Määrittää, siirtyykö komentokielisovellus nauhoitustilaan sen aktivoinnin yhteydessä.

Esimerkkejä komentokielen käytöstä yhteystiedostoissa

Esimerkki 1:

```
[Script]
ScriptEnabled=Yes
ScriptFileName=C:\Shiva\Omatied.Scp
ScriptTerminal=Yes
ScriptRecord=No
```

Example 2:

```
[Script]
ScriptEnabled=Yes
ScriptFileName=C:\Shiva\Uusitied.Scp
ScriptTerminal=Yes
ScriptRecord=Yes
```

Huomautus

- Esimerkin 2 määrittäminen näyttää tapahtumat nauhoituksen aikana.

Komentosarjatiedoston luominen Internet Explorerissa

Ennen komentosarjatiedoston luomista käyttäjälle sinun on nauhoitettava komentosarjatiedosto Scripter.exe-ohjelmalla. Kun olet luonut tiedoston, käyttäjä voi muokata sitä ja lisätä siihen oman käyttäjätunnuksensa ja salasanasensa.

Komentosarjatiedoston nauhoittaminen ja toistaminen TTY-ikkunassa

- 1 Käynnistä Scripter.exe-ohjelma. Se sijaitsee tavallisesti samassa hakemistossa kuin Internet Explorer -ohjelma.
- 2 Valitse **Tiedosto Nauhoita**.
- 3 Etsi komentosarjatiedosto (.scp) ja valitse se tiedostoluettelosta. Voit myös kirjoittaa uuden tiedostonimen. Kun aloitat nauhoittamisen, näyttöön tulee seuraava ikkuna.
- 4 Avaa Internet Explorer -ohjelmaryhmä ja valitse Uusi yhteys. Luo uusi yhteys näyttöön tulevien ohjeiden avulla tai valitse aiemmin luomasi yhteys.
- 5 Valitse yhteyskuvake. Valitse Tiedosto Ominaisuudet. Varmista, että Näytä pääteikkuna valinnan jälkeen -valintaruutu on valittuna.
- 6 Odota, kunnes yhteys Internet-palvelimeen on saatu. ShivaRemote-ohjelman komentoikkunaan tulee kehote "Started".
- 7 Napsauta nauhoitusikkunaa (joka toimii nyt pääteikkunana), ja kirjoita sitten tiedot, joita tarvitaan kirjaututtaessa Internet-palvelutoimittajan palvelimelle.
- 8 Kun kirjautuminen on onnistunut, valitse komentosarjaohjelman ikkunasta Lopeta nauhoitus.
- 9 Valitse ShivaPPP:n komentosarjaohjelman ikkunasta OK.
- 10 Voit lopettaa tietoliikenneyhteyden valitsemalla yhteysikkunasta Peruuta.

Huomautus

Käyttäessään tätä komentosarjatiedostoa käyttäjän kannattaa ehkä poistaa Näytä pääteikkuna valinnan jälkeen -valintaruudun valinta.

Kun olet kirjoittanut komentosarjatiedostoon tarvittavat tiedot, valitse Stop. Ikkuna sulkeutuu ja normaali PPP-tiedonsiirto alkaa.

Ohjelman luoma komentosarjatiedosto on seuraavanlainen:

```
; C:\SCRIPTS\IE\LOGIN.SCP
; Created: 7-1-1996 at 17:07:11
;
;
proc main
string szPassword
transmit "^M"
waitfor "Host Name: ",matchcase until 15
transmit "spsy01^M"
waitfor "UIC: ",matchcase until 6
transmit "spsy053514^M"
waitfor "Password: ",matchcase until 3
if $PASSWORD then
transmit $PASSWORD
transmit "^M"
goto doneTxPassword
endif
getinput "Password: " szPassword
transmit szPassword
transmit "^M"
doneTxPassword:
endproc
```

Nauhoitetun komentosarjatiedoston muokkaaminen tai komentosarjatiedoston kirjoittaminen manuaalisesti

Scripter.exe-ohjelma nauhoittaa komentosarjatiedoston, joka sisältää yksilöityjä käyttäjäkohtaisia tietoja, esimerkiksi käyttäjänimen ja salasanan. Jos olet palvelutoimittaja, haluat ehkä luoda yleisen komentosarjatiedoston, josta kukin käyttäjä voi muokata omien tietojensa mukaisen version.

Seuraavan esimerkkitiedoston avulla voit luoda yleisen komentosarjatiedoston, joka ei sisällä yksilöityjä käyttäjätietoja.

```
;-----  
; Tiedostonimi: LOGIN01.SCP  
; Luontipvm:    18.10.1996  
; Luontiaika:   12:00:00  
;-----  
  
proc main  
  
; Jotkin järjestelmät aloittavat kirjautumisen  
; näppäinpainalluksella  
;-----  
-----  
  
transmit "^M"  
  
; Lähetä palvelutoimittajan määrittämä pääkoneen nimi  
;-----  
  
waitfor "Host Name:"  
transmit "hostname"  
transmit "^M"  
  
; Lähetä käyttäjänimi,  
; joka saadaan vastaavan yhteystiedoston Käyttäjänimi-  
kentästä.  
;-----  
-  
  
waitfor "Username:"  
transmit $USERID  
transmit "^M"  
  
; Lähetä salasana,  
; joka saadaan vastaavan yhteystiedoston Salasana-  
kentästä.  
;-----  
  
waitfor "Password:"  
transmit $PASSWORD  
transmit "^M"
```

endproc

Lisätietoja komentosarjojen käytöstä

1.0 Johdanto

Monilla Internet-palvelutoimittajilla ja useissa online-palveluissa yhteyden muodostaminen edellyttää käyttäjänimen, salasanan ja muiden vastaavien tietojen kirjoittamista. Puhelinyhteyden komentokieltä käyttämällä nämä toiminnot voidaan automatisoida.

Komentosarjatiedosto on tekstitiedosto, joka sisältää komentoja, parametreja ja lausekkeita. Niiden avulla ohjelma luo yhteyden Internet-palvelutoimittajaan tai online-palveluun, jolloin käyttäjä voi käyttää kyseisiä palveluja. Voit luoda oman komentosarjatiedoston millä tahansa tekstinkäsittelyohjelmalla, esimerkiksi Microsoftin Muistiolla. Kun olet luonut komentosarjatiedoston, voit liittää siihen tietyn puhelinyhteyden komentosarjatiedoston Puhelinyhteyksien komentosarjatyökalulla.

2.0 Komentosarjan perusrakenne

Komentosarjatiedosto rakentuu komennoista. Jotkin komennot tarvitsevat parametreja tarkentamaan komennon tehtävää. Lauseke on toimintojen ja argumenttien muodostama kokonaisuus, joka antaa tulokseksi arvon. Lausekkeita voi käyttää minkä tahansa komennon arvoina. Niitä ovat esimerkiksi aritmeettiset lausekkeet, vertailulausekkeet ja merkkijonojen liittämislausekkeet.

Puhelinyhteyden komentosarjan perusmuoto on seuraava:

```
;
; Kommentti alkaa puolipisteellä ja päättyy
; rivin loppuun.
;

proc main
    ; Komentosarja voi sisältää muuttujia
    ; ja komentoja.

    muuttujien määrittelyt

    komennot

endproc
```

Komentosarjassa on oltava main-toimintosarja, jonka alku määritetään **proc**-avainsanalla ja loppu **endproc**-avainsanalla.

Muuttujat on määriteltävä ennen niiden käyttämistä komennoissa. Ohjelma suorittaa ensin main-toimintosarjan ensimmäisen komennon ja sen jälkeen muut komennot siinä järjestyksessä, jossa ne esiintyvät komentosarjassa. Komentosarja loppuu, kun ohjelma saavuttaa main-toimintosarjan loppukomennon.

3.0 Muuttujat

Komentosarjat voivat sisältää muuttujia. Muuttujan nimen alussa on oltava kirjain tai alaviiva (`_`), ja se voi sisältää määrittämättömän määrän isoja tai pieniä kirjaimia, numeroita ja alaviivoja. Varattua sanaa ei voi käyttää muuttujan nimenä. Varatut sanat on lueteltu tämän tiedoston lopussa.

Muuttujat on määriteltävä ennen niiden käyttämistä. Muuttujan määrittelemisen yhteydessä määritetään myös sen laji. Tietyn lajinen muuttuja voi saada vain samaa lajia olevia arvoja. Komentosarjatyökalu tukee seuraavia kolmea muuttujalajia:

Laji Integer
Kuvaus Negatiivinen tai positiivinen kokonaisluku, esimerkiksi 7, -12 tai 5698.

Laji String
Kuvaus Lainausmerkkeihin kirjoitettu merkkijono, esimerkiksi "Hei kaikki!" tai "Kirjoita salasana:".

Laji Boolean
Kuvaus Looginen totuusarvo TRUE (tosi) tai FALSE (epätosi).

Muuttujalle annetaan arvo seuraavan määrittyslauseen avulla:

```
muuttuja = lauseke
```

Muuttuja saa lasketun arvon.

Esimerkkejä:

```
integer luku = 5
integer aikakatk = (4 * 3)
integer i

boolean bValmis = FALSE

string szIP = (getip 2)

set iposoite szIP
```

Edellä olevissa esimerkeissä on käytetty komentoja, jotka eivät ole Shivan tukemia.

3.1 Järjestelmämuuttujat

Järjestelmämuuttujat ovat komentokielen määrittämiä tai määräytyvät yhteyden muodostamisessa annettujen tietojen mukaan. Ne ovat vain luku -muuttujia, eli niitä ei voi muuttaa komentosarjan sisällä. Järjestelmämuuttujat ovat seuraavat:

Nimi \$USERID
Tyyppi String
Kuvaus Käyttäjän tunnistustiedot nykyisessä yhteydessä. Tämä muuttuja saa arvonsa yhteysohjelman Yhdistä-valintaruutuun määritettyjen käyttäjätietojen perusteella.

Nimi \$PASSWORD_
Tyyppi String
Kuvaus Nykyisen yhteyden salasana. Tämä muuttuja saa arvonsa yhteysohjelman Yhdistä-valintaruutuun määritettyjen käyttäjätietojen perusteella.

Nimi \$SUCCESS
Tyyppi Boolean
Kuvaus Tämä muuttuja saa arvon, kun eräät komennot lähettävät tiedon komennon mahdollisesta onnistumisesta. Komentosarjaan voidaan määrittää ehtolauseita tämän tiedon perusteella.

Nimi \$FAILURE
Tyyppi Boolean
Kuvaus Tämä muuttuja saa arvon, kun eräät komennot lähettävät tiedon komennon mahdollisesta epäonnistumisesta. Komentosarjaan voidaan määrittää ehtolauseita tämän tiedon perusteella.

Näitä muuttujia voidaan käyttää, kun käytettävä lauseke on samaa tyyppiä. Esimerkiksi

```
transmit $USERID
```

ei ole kelvollinen komento, koska \$USERID on merkkijonotyyppinen muuttuja.

4.0 Merkkijonoliteraalit

Puhelinyhteyden komentosarjat tukevat Escape-koodeja sekä ^-merkkijonoja seuraavalla tavalla:

<u>Merkkijonoliteraali</u>	<u>Kuvaus</u>
<u>Merkkijonoliteraali</u>	<u>^merkki</u>
<u>Kuvaus</u>	Merkin määritys sirkumfleksin avulla

Jos *merkki* on arvojen '@' ja '_' välillä, merkkijonoliteraali tulkitaan yksitavuisesti arvoksi väliltä 0 - 31. Esimerkiksi ^M tulkitaan palautukseksi rivin alkuun.

Jos *merkki* on arvojen 'a' ja 'z' välillä, merkkijonoliteraali tulkitaan yksitavuisesti arvoksi väliltä 1 - 26.

Jos *merkki* on jokin muu arvo, merkkijonoliteraalia ei käsitellä erikoismerkinä.

<u>Merkkijonoliteraali</u>	<u><cr></u>
<u>Kuvaus</u>	Palautus rivin alkuun.

<u>Merkkijonoliteraali</u>	<u><lf></u>
<u>Kuvaus</u>	Rivinsiirto.

<u>Merkkijonoliteraali</u>	<u>\"</u>
<u>Kuvaus</u>	Lainausmerkki.

<u>Merkkijonoliteraali</u>	<u>\^</u>
<u>Kuvaus</u>	Sirkumfleksi.

<u>Merkkijonoliteraali</u>	<u>\<</u>
<u>Kuvaus</u>	Pienempi kuin -merkki (<).

<u>Merkkijonoliteraali</u>	<u>\\</u>
<u>Kuvaus</u>	Kenoviiva.

Esimerkkejä:

```
transmit "^M"  
transmit "Jussi^M"  
transmit "<cr><lf>"  
waitfor "<cr><lf>"
```

5.0 Lausekkeet

Lauseke on operaattoreiden ja argumenttien yhdistelmä, joka antaa tulokseksi arvon. Lausekkeita voidaan käyttää minkä tahansa komennon arvoina.

Lauseke voi sisältää muuttujia, kokonaislukuarvoja, merkkijonoarvoja, totuusarvoja ja mitä tahansa seuraavissa taulukoissa luetelluista yksinkertaisista operaattoreista ja binaarioperaattoreista. Yksinkertaiset operaattorit ovat suoritussuunnassa ensisijaisia. Binaarioperaattoreiden suoritussuunnassa on sama kuin niiden sijainti taulukossa.

Yksinkertaiset operaattorit:

<u>Operaattori</u>	-
<u>Operaation tyyppi</u>	Miinus
<u>Operaattori</u>	!
<u>Operaation tyyppi</u>	Yhden komplementti

Binaarioperaattorit on lueteltu seuraavassa taulukossa niiden suoritusjärjestyksessä. Ensin suoritettavat operaattorit ovat luettelossa ensimmäisenä.

<u>Operaattorit</u>	* /
<u>Operaation tyyppi</u>	Kertolasku
<u>Tyyppirajoitukset</u>	Kokonaisluvut
<u>Operaattorit</u>	+ -
<u>Operaation tyyppi</u>	Yhteenlasku kokonaisluvut
<u>Tyyppirajoitukset</u>	Merkkijonot (vain +)
<u>Operaattorit</u>	< > <= >=
<u>Operaation tyyppi</u>	Vertailu
<u>Tyyppirajoitukset</u>	Kokonaisluvut
<u>Operaattorit</u>	== !=
<u>Operaation tyyppi</u>	Yhtäsuuruus
<u>Tyyppirajoitukset</u>	Kokonaisluvut, merkkijonot, totuusarvot
<u>Operaattorit</u>	and
<u>Operaation tyyppi</u>	Looginen JA
<u>Tyyppirajoitukset</u>	Totuusarvot
<u>Operaattorit</u>	or
<u>Operaation tyyppi</u>	Looginen TAI
<u>Tyyppirajoitukset</u>	Totuusarvot

Esimerkkejä:

```
luku = 3 + 5 * 40
transmit "Hei" + " kaikki"
delay 24 / (7 - 1)
```

6.0 Kommentit

Puolipisteen jäljessä olevaa tekstiä ei pidetä komentokieleen kuuluvana.

Esimerkkejä:

```
; Tämä on kommentti.

transmit "Hei" ; lähettää merkkijonon "hei".
```

7.0 Avainsanat

Avainsanat määrittävät komentokielen rakenteen. Toisin kuin komennot, ne eivät suorita toimintoa. Avainsanat on lueteltu seuraavassa.

proc nimi

Osoittaa toimintasarjan alkukohtaan. Kaikissa komentosarjoissa on oltava pää- eli main-toimintasarja (**proc main**). Kommentosarjan suorittaminen alkaa main-toimintasarjan alusta ja päättyy sen lopussa.

endproc

Osoittaa toimintasarjan loppukohtaan. Kun komentosarjan suoritus on edennyt main-

toimintasarjan **endproc**-lauseeseen, Puhelinyhteyden komentotyökalu käynnistää PPP- tai SLIP-toiminnon.

integer *nimi* [= *arvo*]

Määrittelee kokonaislukumuuttujan. Muuttujan alkuarvoksi voi määrittää minkä tahansa numeerisen arvon tai muuttujan.

string *nimi* [= *arvo*]

Määrittelee merkkijonomuuttujan. Muuttujan alkuarvoksi voi määrittää minkä tahansa merkkijonoliteraalin tai muuttujan.

boolean *nimi* [= *arvo*]

Määrittelee totuusarvomuttujan. Muuttujan alkuarvoksi voi määrittää minkä tahansa totuusarvon tai muuttujan.

8.0 Komennot

Internet Explorerin Numeronvalitsin-ohjelman käyttämä komentokieli kuuluu osana Microsoft Windows(R) 95:n määrittämään Puhelinyhteyden komentokieleen.

Kaikki komennot ovat varattuja sanoja, mikä tarkoittaa, että komentosarjoissa ei voi määrittää muuttujia, joilla on sama nimi kuin komennolla. Seuraavassa on luettelo komennosta:

delay *sekunnit*

Pitää ennen seuraavan komennon suorittamista tauon, jonka pituus määräytyy argumentin *sekunnit* mukaan.

Esimerkkejä:

```
delay 2      ; Pitää 2 sekunnin tauon.  
delay x * 3  ; Pitää x * 3 sekunnin tauon.
```

goto *otsikko*

Siirtyy komentosarjassa argumentin *otsikko* määrittämään kohtaan ja jatkaa komentojen suorittamista tuosta kohdasta.

Esimerkki:

```
waitfor "Prompt>" until 10  
if !$SUCCESS then  
    goto Onnistui; Siirtyy kohtaan Onnistui ja suorittaa  
    ; sitä seuraavat komennot.  
endif  
  
transmit "bbs^M"  
goto End  
  
Onnistui:  
transmit "^M"
```

halt

Pysäyttää komentosarjan suorittamisen. Tämä komento ei poista päätteikkunaa näytöstä. Yhteys muodostetaan valitsemalla Jatka-painike. Komentosarjaa ei voi aloittaa uudelleen.


```
if ehto then
    komennot
endif
```

Suorittaa määritetyt *komennot*, jos *ehto* on TOSI.

Esimerkki:

```
if $USERID == "Jussi" then
    transmit "Jussi^M"
endif
```

otsikko :

Määrittää komentosarjan kohdan, johon ohjelman suorituksen on siirryttävä. Otsikon on oltava yksilöllinen, ja sen on noudatettava muuttujien nimeämissääntöjä.

```
transmit merkkijono [ , raw ]
```

Määrittää etätietokoneelle argumentin *merkkijono* määrittämät merkit.

Etätietokone tulkitsee Escape-koodit ja ^-merkkijonot, ellei lisä komentoon parametria **raw**. Tämän parametrin käyttäminen on tarpeen silloin, kun lähetetään \$USERID- sekä \$PASSWORD-järjestelmämuuttujia ja käyttäjänimi tai salasana sisältää merkkejä, jotka ohjelma tulkitseisi muuten Escape-koodiksi tai ^-merkkijonoiksi.

Esimerkkejä:

```
transmit "slip" + "^M"
transmit $USERID, raw
```

```
waitfor merkkijono [ , matchcase ] [ then otsikko
{ , merkkijono [ , matchcase ] then otsikko } ]
[ until aika ]
```

Odottaa, kunnes tietokoneesi vastaanottaa etätietokoneelta yhden tai useita määritetyistä merkkijonoista. *Merkkijono*-parametri ei ota kirjainkoko huomioon, ellei lisä komentoon **matchcase**-parametria.

Jos vastaava merkkijono vastaanotetaan, ja ohjelmassa käytetään **then** *otsikko* -parametria, tämä komento siirtää ohjelman suorituskohdan argumentin *otsikko* osoittamaan kohtaan.

Valinnainen **until** *aika* -parametri määrittää sen enimmäissekuntimäärän sekunteina, jonka ajan tietokone odottaa merkkijonon vastaanottamista ennen seuraavan komennon suorittamista. Jos tätä parametria ei määritetä, tietokone jää odotustilaan.

Jos tietokone vastaanottaa jonkin määritetyistä merkkijonoista, järjestelmämuuttuja \$SUCCESS saa arvon TOSI. Se saa arvon EPÄTOSI, jos argumentin *aika* määrittämä aika on kulunut umpeen eikä merkkijonoa ole vastaanotettu.

Esimerkkejä:

```
waitfor "Login:"
waitfor "Password?", matchcase
```

```

waitfor "prompt>" until 10

waitfor
    "Login:"           then TeeKirjautuminen,
    "Password:"        then TeeSalasana,
    "BBS:"             then TeeBBS,
    "Other:"           then TeeMuu
until 10

```

Seuraava komento on Shiva-laajennusosan komento, jota voidaan käyttää Microsoftin komentokielen komentosarjoissa.

getinput "Näyttömerkkijono " szToReceiveInput

Kysyy käyttäjältä tietoja komentosarjan suorittamisen aikana.

Esimerkki:

```
Getinput "Järjestelmän salasana" szSystemPassword
```

9.0 Varatut sanat

Seuraavat sanat on varattu, eikä niitä voida käyttää muuttujien niminä:

and	boolean	databits	delay
do	endif	endproc	endwhile
even	FALSE	getip	goto
halt	if	integer	ipaddr
keyboard	mark	matchcase	none
odd	off	on	or
parity	port	proc	raw
screen	set	space	stopbits
string	then	transmit	TRUE
until	waitfor	while	

10.0 Shivan komentosanat, joita komentokieli ei tue

```

getip
port databits
port parity
port stopbits
set screen keyboard
ipaddr
while/endwhile

```